
Recommandation UIT-T G.107 
 

Le modèle E, modèle de calcul utilisé pour la planification de la transmission 
 

 

 

Résumé 
La présente Recommandation UIT-T décrit l'algorithme du modèle dit modèle E en tant que modèle 
UIT-T commun d'évaluation de l'indice de transmission. Ce modèle de calcul peut être utile à ceux 
qui planifient la transmission pour leur permettre de s'assurer que les usagers sont satisfaits de la 
qualité de la transmission de bout en bout. Le résultat principal fourni par ce modèle est une 
évaluation scalaire de la qualité de la transmission. Une propriété fondamentale de ce modèle est 
l'utilisation de facteurs de dégradation de la transmission qui rendent compte des effets des 
équipements modernes de traitement des signaux.  

La présente révision donne une version améliorée du modèle E pour mieux tenir compte des effets 
du bruit de salle du côté émission et de la distorsion de quantification. 

 

 

Source 
La Recommandation G.107 de l'UIT-T, révisée par la Commission d'études 12 (1997-2000) de 
l'UIT-T, a été approuvée le 18 mai 2000 selon la procédure définie dans la Résolution 1 de la 
CMNT. 

AVANT-PROPOS 

L'UIT (Union internationale des télécommunications) est une institution spécialisée des Nations Unies dans le 
domaine des télécommunications. L’UIT-T (Secteur de la normalisation des télécommunications) est un 
organe permanent de l'UIT. Il est chargé de l'étude des questions techniques, d'exploitation et de tarification, 
et émet à ce sujet des Recommandations en vue de la normalisation des télécommunications à l'échelle 
mondiale. 

La Conférence mondiale de normalisation des télécommunications (CMNT), qui se réunit tous les quatre ans, 
détermine les thèmes d'études à traiter par les Commissions d'études de l'UIT-T, lesquelles élaborent en retour 
des Recommandations sur ces thèmes. 

L'approbation des Recommandations par les Membres de l'UIT-T s'effectue selon la procédure définie dans la 
Résolution 1 de la CMNT. 

Dans certains secteurs des technologies de l'information qui correspondent à la sphère de compétence de 
l'UIT-T, les normes nécessaires se préparent en collaboration avec l'ISO et la CEI. 

 

 

 

 

 

NOTE 

Dans la présente Recommandation, l'expression "Administration" est utilisée pour désigner de façon abrégée 
aussi bien une administration de télécommunications qu'une exploitation reconnue. 

 

 

 

 

DROITS DE PROPRIÉTÉ INTELLECTUELLE 

L'UIT attire l'attention sur la possibilité que l'application ou la mise en œuvre de la présente Recommandation 
puisse donner lieu à l'utilisation d'un droit de propriété intellectuelle. L'UIT ne prend pas position en ce qui 
concerne l'existence, la validité ou l'applicabilité des droits de propriété intellectuelle, qu'ils soient 
revendiqués par un Membre de l'UIT ou par une tierce partie étrangère à la procédure d'élaboration des 
Recommandations. 

A la date d'approbation de la présente Recommandation, l'UIT n'avait pas été avisée de l'existence d'une 
propriété intellectuelle protégée par des brevets à acquérir pour mettre en œuvre la présente Recommandation. 
Toutefois, comme il ne s'agit peut-être pas de renseignements les plus récents, il est vivement recommandé 
aux responsables de la mise en œuvre de consulter la base de données des brevets du TSB. 

Recommandation UIT-T G.107 

Le modèle E, modèle de calcul utilisé pour la planification de la transmission 

1 Généralités 

1.1 Domaine d'application 
La présente Recommandation UIT-T décrit un modèle de calcul connu sous le nom de modèle E, qui 
s'est avéré utile en tant qu'outil de planification de la transmission pour évaluer les effets conjugués 
des variations des différents paramètres de transmission altérant la qualité en conversation1 de la 
téléphonie à 3,1 kHz avec combiné. Ce modèle de calcul peut être employé, par exemple, par ceux 
qui planifient la transmission pour leur permettre de s'assurer que les usagers sont satisfaits de la 
qualité de la transmission de bout en bout, tout en évitant une ingénierie des réseaux trop complexe. 
Il faut souligner que le résultat brut fourni par ce modèle est le "facteur d'évaluation" R, qui peut être 
transformé pour obtenir des estimations de l'opinion des usagers. De telles estimations sont faites 
seulement à des fins de planification de la transmission et non à des fins de prédiction concrète de 
l'opinion des usagers (pour laquelle il n'existe aucun modèle agréé recommandé par l'UIT-T). En 
conséquence, le modèle décrit ci-après est destiné à être utilisé pour effectuer des comparaisons 
relatives de conditions de transmission. 

La présente révision donne une version améliorée du modèle E pour mieux tenir compte des effets 
du bruit de salle du côté émission et de la distorsion de quantification. Les utilisateurs devraient être 
informés du fait qu'en conséquence l'algorithme de calcul a été légèrement modifié (voir 
l'Annexe C). 

Le modèle E n'a pas été complètement vérifié au moyen de sondages sur le terrain ou d'essais en 
laboratoire pour les très nombreuses combinaisons possibles de paramètres d'entrée. Pour beaucoup 
de combinaisons très importantes aux yeux de ceux qui planifient les transmissions, le modèle E peut 
être utilisé avec confiance, mais pour d'autres combinaisons de paramètres, les prédictions du 
modèle E ont été mises en doute. Elles sont actuellement à l'étude. Il faut par conséquent prendre des 
précautions en employant le modèle E dans certaines situations, celui-ci pouvant par exemple donner 
des résultats inexacts pour la combinaison de certains types de dégradation. L'Annexe A donne de 
plus amples renseignements à ce sujet.  

1.2 Références normatives 
La présente Recommandation se réfère à certaines dispositions des Recommandations UIT-T et 
textes suivants qui, de ce fait, en sont partie intégrante. Les versions indiquées étaient en vigueur au 
moment de la publication de la présente Recommandation. Toute Recommandation ou tout texte 
étant sujet à révision, les utilisateurs de la présente Recommandation sont invités à se reporter, si 
possible, aux versions les plus récentes des références normatives suivantes. La liste des 
Recommandations de l'UIT-T en vigueur est régulièrement publiée. 

[1] Recommandation UIT-T G.100 (1993), Définitions utilisées dans les Recommandations sur 
les caractéristiques générales des connexions et des circuits téléphoniques internationaux. 

[2] Recommandation UIT-T G.108 (1999), Application du modèle E: guide de planification. 

____________________ 
1  La qualité en conversation, dans ce contexte, concerne les caractéristiques de transmission, par exemple les 

longs temps de transmission, les effets des échos pour le locuteur, etc. Cependant le modèle E décrit dans 
la présente Recommandation UIT-T n'est pas conçu pour modifier les dégradations de transmission durant 
les périodes de parole simultanée. 

[3] Recommandation UIT-T G.109 (1999), Définition de catégories de qualité de transmission 
vocale. 

[4] Recommandation UIT-T G.113 (1996), Dégradations de la transmission. 

[5] Recommandation G.113 – Appendice I (1999), Dégradations de la transmission – 
Appendice I: Valeurs de planification provisoires pour le facteur de dégradation due à 
l'équipement. 

2 Le modèle E, modèle de calcul destiné à être employé en planification de la 
transmission 

2.1 Introduction 
En planification de la transmission, la complexité des réseaux modernes exige que les nombreux 
paramètres de transmission ne soient pas seulement examinés individuellement, mais qu'il soit aussi 
tenu compte des effets de leur combinaison. Bien que cela puisse se faire à l'aide d'une "estimation 
experte et documentée", une démarche plus systématique, telle que celle qui consiste à employer un 
modèle de calcul, est préférable. Le résultat du modèle décrit ici est une valeur scalaire R mesurant 
la qualité de transmission, et variant proportionnellement à la qualité globale en conversation. La 
Recommandation UIT-T G.113 [4] donne des indications concernant certaines dégradations, 
notamment sur leurs effets conjugués, sur la base d'une version simplifiée du modèle. Toutefois, les 
résultats peuvent aussi consister en des estimations théoriques concernant les réactions des usagers, 
par exemple sous la forme de pourcentages indiquant que la connexion modélisée est "au moins 
bonne" ou "au mieux médiocre". Cela est décrit à l'Annexe B. Par ailleurs, la 
Recommandation UIT-T G.108 [2] donne des indications détaillées sur l'application adéquate du 
modèle E, tel qu'il est décrit dans la présente Recommandation UIT-T, et la 
Recommandation UIT-T G.109 [3] contient la définition des catégories de qualité de transmission 
vocale. 

2.2 Code source  
L'Annexe C contient le code source, en langage BASIC, du modèle E décrit dans la présente 
Recommandation UIT-T. La fonction de ce code est de faire en sorte que les utilisateurs du 
modèle E appliquent les formules de façon uniforme. 

3 Structure et algorithmes fondamentaux du modèle E 
Le modèle E, fondé sur la méthode du facteur de dégradation due à l'équipement, fait suite à des 
modèles antérieurs d'évaluation de l'indice de transmission. Il a été élaboré par un groupe ad hoc de 
l'ETSI appelé "Voice Transmission Quality from Mouth to Ear" (qualité de transmission vocale de la 
bouche à l'oreille).  

La connexion de référence montrée dans la Figure 1, est divisée en deux parties: le côté émission et 
le côté réception. Le modèle a pour but d'évaluer la qualité en conversation de la bouche à l'oreille, 
perçue par l'usager du côté réception à la fois en tant qu'auditeur et en tant que locuteur. 

 le locuteur TELR

 

Figure 1/G.107 −−−− Connexion de référence du modèle E 

Les paramètres de transmission utilisés comme entrées du modèle de calcul sont représentés dans la 
Figure 1. Les valeurs du bruit de salle et des facteurs D sont traitées séparément dans l'algorithme 
pour le côté émission et pour le côté réception. Elles peuvent être de grandeur différente. Les 
paramètres SLR, RLR et le bruit de circuit Nc sont établis par rapport à un point de référence défini 
à 0 dBr. Tous les autres paramètres d'entrée sont considérés comme des valeurs s'appliquant à 
l'ensemble de la connexion, telles que l'équivalent OLR (qui est toujours la somme de SLR et 
de RLR), le nombre d'unités qdu, les facteurs de dégradation due à l'équipement Ie et le facteur 
d'avantage A, ou se réfèrent uniquement au côté réception, telles les valeurs STMR, LSTR, WEPL 
(pour le calcul de l'écho pour l'auditeur) et TELR. 

Trois paramètres différents sont associés au temps de propagation. Le temps de propagation absolu 
Ta représente le temps total dans un sens entre les côtés émetteur et récepteur; il est utilisé pour 
évaluer les dégradations dues à des temps de propagation excessivement longs. Le temps de 
propagation moyen dans un sens T est le temps qui s'écoule entre le côté réception (en état 
d'émission) et le point d'une connexion où un couplage de signaux donne naissance à un écho. Le 
temps de propagation aller-retour Tr représente le temps dans une boucle à quatre fils où le signal 
"doublement réfléchi" entraîne des dégradations dues à l'écho pour l'auditeur. 

3.1 Calcul du facteur d'évaluation de l'indice de transmission R 
En accord avec la méthode du facteur de dégradation due à l'équipement, le principe fondamental du 
modèle E repose sur un concept issu de la description du modèle OPINE [voir la bibliographie, 
Supplément 3 aux Recommandations de la série P]: 

Les facteurs psychologiques définis sur l'échelle psychologique sont additifs. 

Le résultat brut de tout calcul fait au moyen du modèle E est le facteur d'évaluation de l'indice de 
transmission R, qui combine tous les paramètres de transmission s'appliquant à la connexion. Ce 
facteur d'évaluation R est composé de: 

  AIeIdIsRoR +−−−=  (1) 

Ro représente en principe le rapport signal/bruit de base, et inclut l'effet des sources de bruit telles 
que le bruit de circuit et le bruit de salle. Le facteur Is est une combinaison de toutes les dégradations 
qui surviennent plus ou moins simultanément avec le signal vocal. Le facteur Id représente les 
dégradations causées par le temps de propagation; le facteur de dégradation due à l'équipement, Ie, 
représente les dégradations causées par les codecs à faible débit. Le facteur d'avantage A permet de 
compenser les facteurs de dégradation quand l'accès fourni à l'usager représente un certain bénéfice. 
Le terme Ro ainsi que les valeurs Is et Id sont subdivisés en d'autres valeurs de dégradation 
spécifiques. Les sous-paragraphes suivants contiennent les formules utilisées dans le modèle E. 

3.2 rapport signal/bruit de base Ro 
Le rapport signal/bruit de base Ro est défini par: 

  ( )NoSLRRo +−= 5,115  (2) 

Le terme No [en dBm0p] est la sommation en puissance des différentes sources de bruit: 

  
�
�

�

�

�
�

�

�
+++= 10101010 10101010lg10

NfoNorNosNc

No  (3) 

Nc [en dBm0p] est la somme de toutes les puissances de bruit du circuit, chacune par rapport au 
point de référence 0 dBr. 

Nos [en dBm0p] est le bruit de circuit équivalent au point 0 dBr dû au bruit de salle Ps du côté 
émission: 

  ( )214004,0100 −−−+−−−= DsOLRPsDsSLRPsNos  (4) 

où OLR = SLR + RLR. D'une manière analogue, le bruit de salle Pr du côté réception conduit à un 
bruit de circuit équivalent Nor [en dBm0p] au point 0 dBr. 

  2)35(008,0121 −++−= PrePreRLRNor  (5) 

Le terme Pre [en dBm0p] est le "bruit de salle effectif" dû à la transformation du bruit Pr par le trajet 
d'effet local pour l'auditeur: 

  
�
�

�

�

�
�

�

�
++= 10

)–10(

101lg10
LSTR

PrPre  (6) 

Nfo [en dBm0p] représente le "bruit de fond" du côté réception, 

  RLRNforNfo +=  (7) 

Nfor étant généralement choisi égal à –64 dBmp. 

3.3 facteur de dégradation simultanée Is 
Le facteur Is est la somme de toutes les dégradations pouvant survenir plus ou moins en même temps 
que la transmission vocale. Il est subdivisé en trois facteurs de dégradation spécifiques: 

  IqIstIolrIs ++=  (8) 

Le facteur Iolr représente la diminution de la qualité causée par des valeurs trop faibles d'OLR; il est 
donné par: 

où: 

  )64(2,0 RLRNoOLRX −++=  (10) 

Le facteur Ist représente la dégradation causée par un effet local n'atteignant pas une valeur optimale: 

où: 

10410 10e10lg10–

Le facteur de dégradation Iq représente la dégradation causée par la distorsion de quantification: 

  [ ]ZYIq 10101lg15 ++=  (13) 

où: 

  
94,8

46
15

100 GR
Y o −+

−
=  (14) 

  
4030

46 GZ −=  (15) 

et: 

  20602,0258,007,1 QQG ++=  (16) 

  )qdulg(1537 −=Q  (17) 

Dans cette formule, le terme qdu représente le nombre d'unités qdu pour l'ensemble de la connexion 
entre le côté émission et le côté réception. 
NOTE – Si l'on utilise le facteur de dégradation Ie pour un équipement, il ne faut pas utiliser la valeur qdu 
pour ce même équipement. 

3.4 facteur de dégradation due au temps de propagation Id 
Le facteur de dégradation Id représente toutes les dégradations dues au temps de propagation des 
signaux vocaux; il est lui aussi subdivisé en trois facteurs, à savoir Idte, Idle et Idd: 

  IddIdleIdteId ++=  (18) 

Le facteur Idte donne une estimation des dégradations dues à l'écho pour le locuteur: 

  ( )TReRoeReRoeIdte –
2

e–11–100
4

2
–

�

++=  (19) 

où: 

  )(5,1 RLRNoRoe −−=  (20) 

  )14(5,280 −+= TERVRe  (21) 

  
23,0–e6

150
1

10
1

lg40– T
T

T

TELRTERV +
+

+
=  (22) 

Pour des valeurs de T < 1 ms, l'écho pour le locuteur doit être considéré comme un effet local, 
autrement dit Idte = 0. L'algorithme de calcul combine par ailleurs l'influence de STMR et de l'écho 
pour le locuteur. Si l'on tient compte du fait que des faibles valeurs d'affaiblissement STMR peuvent 
avoir certains effets de masquage sur l'écho pour le locuteur et que pour les valeurs très élevées de 
l'affaiblissement STMR l'écho pour le locuteur peut devenir plus perceptible, les termes TERV et Idte 
sont ajustés de la manière suivante: 

Pour STMR < 9 dB: 

Dans l'équation (20) TERV est remplacé par TERVs, où: 

  
2

IstTERVTERVs +=  (23) 

Pour 9 dB ≤ STMR ≤ 15 dB: 
les équations (18) à (21) s'appliquent. 

Pour STMR > 15 dB: 

Idte est remplacé par Idtes, où: 

  22 IstIdteIdtes +=  (24) 

Le facteur Idle représente les dégradations dues à l'écho pour l'auditeur. Les équations sont: 

  169
4

)–(
2
– 2

++= RleRoRleRoIdle  (25) 

où: 

  25,0–)1)(7(5,10 ++= TrWEPLRle  (26) 

Le facteur Idd représente la dégradation provoquée par un temps de propagation absolu Ta excessif, 
qui se produit même en cas d'annulation parfaite de l'écho. 

Pour Ta < 100 ms: 

  0=Idd  

Pour Ta > 100 ms: 

avec: 

3.5 Facteur de dégradation due à l'équipement Ie 
Les valeurs du facteur de dégradation Ie des éléments utilisant des codecs à faible débit ne sont pas 
liées aux autres paramètres d'entrée. Elles proviennent de résultats sous formes de notes moyennes 
d'opinion issues d'essais subjectifs et de l'expérience du réseau. Prière de consulter 
l'Appendice I/G.113 [5] pour savoir exactement quelles sont les valeurs recommandées pour Ie. 

3.6 Facteur d'avantage A 
En raison de la signification particulière du facteur d'avantage A, celui-ci n'a pas de relation avec les 
autres paramètres de transmission. Certaines valeurs provisoires sont données dans le Tableau 1. 

Tableau 1/G.107 −−−− Exemples provisoires du facteur d'avantage A 

Exemple de système de communication Valeur maximale de A 

Classique (système filaire) 0 
Mobile par réseau cellulaire dans un bâtiment 5 
Mobile dans une zone géographique ou en déplacement dans un 
véhicule 

10 

Accès à des emplacements difficiles à atteindre, par exemple via des 
connexions par satellite à plusieurs bonds 

20 

Il faut noter que les valeurs du Tableau 1, tirées de la Recommandation UIT-T G.113 [4], ne sont 
que provisoires. L'utilisation du facteur A et la valeur retenue pour celui-ci dans une application 
donnée seront décidées par le planificateur. Toutefois, il faut considérer que les valeurs du Tableau 1 
sont des limites supérieures absolues de A. 

3.7 Valeurs par défaut 
Les valeurs par défaut de tous les paramètres d'entrée utilisés dans l'algorithme du modèle E sont 
réunies dans le Tableau 2. Il est vivement recommandé d'utiliser ces valeurs par défaut pour tous les 
paramètres que l'on ne fait pas varier au cours des calculs de planification. Si tous les paramètres 
sont mis à leurs valeurs par défaut, le calcul conduit à une qualité très élevée (facteur d'évaluation 
R = 93,2). 

Tableau 2/G.107 −−−− Valeurs par défaut et intervalles permis pour les paramètres 

Paramètre Abr. Unité 
Valeur 

par 
défaut 

Intervalle 
permis Remarque 

Equivalent pour la sonie à l'émission (send 
loudness rating) 

SLR dB +8 0 ... +18 Note 1 

Equivalent pour la sonie à la réception 
(receive loudness rating) 

RLR dB +2 −5 ... +14 Note 1 

Affaiblissement d'effet local par la 
méthode de masquage (sidetone masking 
rating) 

STMR dB 15 10 ... 20 Note 2 

Affaiblissement d'effet local pour 
l'auditeur (listener sidetone rating) 

LSTR dB 18 13 ... 23 Note 2 

Valeur D du côté émission téléphonique Ds – 3 −3 ... +3 Note 2 

Valeur D du côté réception téléphonique Dr – 3 −3 ... +3 Note 2 

Equivalent pour la sonie de l'écho pour le 
locuteur (talker echo loudness rating) 

TELR dB 65 5 ... 65  

Affaiblissement pondéré du trajet d'écho 
(weighted echo path loss) 

WEPL dB 110 5 ... 110  

Temps de propagation moyen dans un sens 
du trajet d'écho 

T msec 0 0 ... 500  

Temps de propagation aller-retour dans 
une boucle à 4 fils 

Tr msec 0 0 ... 1000  

Temps de propagation absolu sur 
connexion exempte d'écho 

Ta msec 0 0 ... 500  

Nombre d'unités de distorsion de 
quantification (quantization distortion 
units) 

qdu – 1 1 ... 14  

Facteur de dégradation due à l'équipement Ie – 0 0 ... 40  

Bruit de circuit par rapport au point de 
référence 0 dBr 

Nc dBm0p −70 −80 ... −40  

Seuil de bruit du côté réception Nfor dBmp −64 – Note 3 

Bruit de salle du côté émission Ps dB(A) 35 35 ... 85  
Bruit de salle du côté réception Pr dB(A) 35 35 ... 85  
Facteur d'avantage A – 0 0 ... 20  
NOTE 1 – Valeurs totales entre le microphone ou le récepteur et le point 0 dBr. 
NOTE 2 – Relation fixe: LSTR = STMR + D. 
NOTE 3 – Actuellement en cours d'étude. 

Le texte révisé de la présente Recommandation UIT-T (version 2000) contient une version améliorée 
de l'algorithme du modèle E (voir l'Annexe A). 

Compte tenu de cette révision, le facteur d'évaluation R, calculé sur la base des valeurs par défaut de 
tous les paramètres, a légèrement changé (il est passé de R = 94,2 à R = 93,2). Toutefois, dans la 
pratique, il convient de considérer ce léger écart comme négligeable aux fins de la planification.  

ANNEXE A 

Condition d'utilisation du modèle E 

NOTE – L'évaluation et l'amélioration de l'algorithme du modèle E fera l'objet d'un complément d'étude 
pendant la période d'études 2001-2004. Les résultats obtenus seront inclus dès qu'ils seront disponibles. 

Partie I: Exemples de situations où le modèle E doit être employé avec précaution 

– Niveau global des facteurs de dégradation due à l'équipement 
 Certaines études expérimentales indiquent que les facteurs de dégradation due à 

l'équipement sont généralement trop pessimistes; des marges de sécurité cachées peuvent 
donc être incorporées. 

– Propriété d'additivité globale du modèle 
 Dans le modèle E, on suppose que les différents types de dégradations s'additionnent sur 

l'échelle du facteur d'évaluation de l'indice de transmission R. Cette propriété n'a pas été 
vérifiée de façon satisfaisante. En particulier, on dispose de très peu d'études concernant 
l'interaction entre les codecs à faible débit et les autres types de dégradations, par exemple le 
bruit de salle. En outre, les effets d'ordre lors de la mise en cascade de plusieurs codecs à 
faible débit restent incertains. 

– Prise en compte de l'effet local pour le locuteur 
 Certaines expériences ont démontré qu'il n'est pas tenu compte dans le modèle E de certains 

effets de masquage qui se produisent sur l'effet local pour le locuteur, notamment en rapport 
avec le bruit de circuit, le bruit de salle du côté réception et l'écho pour le locuteur à temps 
de propagation court (< 10 ms). 

– Facteur d'avantage A 
 Jusqu'à présent, il n'a pas été déterminé dans quelles conditions les valeurs données pour le 

facteur d'avantage devraient être appliquées. On s'attend à ce que ces valeurs puissent 
dépendre du groupe d'utilisateurs, par exemple, et à ce que les valeurs absolues changeront à 
long terme. 

– Méthode pour l'obtention des facteurs de dégradation due à un nouvel équipement 
 La CE 12 de l'UIT-T étudie actuellement une nouvelle méthode permettant d'obtenir les 

facteurs de dégradation due à l'équipement. 
– Prédictions pour les différents types de bruit de salle et les différentes formes de fréquence 

dans le canal de communication au cours du trajet d'effet local et du trajet d'écho 
 Le modèle E prend en compte l'effet du bruit de salle exprimé seulement sous forme du 

niveau pondéré A. L'opinion réelle sur la qualité de la communication vocale peut même 
dépendre du type et de la perturbation du bruit environnant. Les caractéristiques 
fréquentielles du canal de la communication, du trajet d'effet local et du trajet d'écho ne sont 
pas prises explicitement en considération dans le modèle E, mais seulement implicitement 
dans les équivalents pour la sonie. Elles peuvent toutefois altérer la qualité de transmission 
perçue. 

Partie II: Situations dans lesquelles le modèle E a été amélioré grâce à une mise à jour des 
versions antérieures 
– Effet du bruit de salle du côté émission 
 Avec le présent algorithme amélioré du modèle E (version 2000), il est tenu compte de 

l'effet Lombard (c'est-à-dire du fait que la personne qui parle adapte sa prononciation et le 
volume de sa voix au bruit environnant). Dans la version 1998, les prédictions du modèle E 
étaient trop pessimistes pour des niveaux de bruit de salle Pr élevés. 

– Prédictions pour la distorsion de quantification 
 Dans le modèle E (version 1998), les résultats des essais subjectifs dans les conditions de 

référence valant pour le générateur MNRU étaient souvent plus pessimistes que les 
prédictions du modèle E. Les courbes présentées à la Figure A.1 ont été obtenues à partir de 
la version 1998 et de la révision 2000 du modèle E, tous les autres paramètres étant mis à 
leurs valeurs par défaut. 

Figure A.1/G.107 −−−− Relation entre le nombre de qdu et le facteur 
d'évaluation R du modèle E 

En ce qui concerne l'algorithme légèrement amélioré du modèle E, tel qu'il figure dans la présente 
Recommandation UIT-T, la relation entre le paramètre qdu et le facteur d'évaluation R du modèle E 
a été modifiée pour que l'algorithme corresponde mieux aux résultats des essais subjectifs 
disponibles. 

ANNEXE B 

Mesures de qualité obtenues à partir du facteur d'évaluation 
de l'indice de transmission R 

Le facteur d'évaluation de l'indice de transmission R peut prendre les valeurs 0 à 100, où R = 0 
représente une qualité extrêmement mauvaise et R = 100 une qualité extrêmement bonne. Le modèle 
E fournit une estimation statistique des mesures de qualité. Les taux de connexions "au moins 
bonnes" (GoB, good or better) ou "au mieux médiocres" (PoW, poor or worse) sont tirés du facteur 
R au moyen de la fonction d'erreur de Gauss 

∞π

=
x t

dtxE
–

2
–

2

e
2
1)(  (B-1) 



 

  UIT-T G.107 (05/2000) 11 

Les équations sont les suivantes: 

  %

�

�= REGoB  (B-2) 

  %
16
–45100 �

�

�
�
�

�= REPoW  (B-3) 

La note moyenne d'opinion (MOS, mean opinion score) sur l'échelle de 1 à 5 peut être obtenue à 
partir du facteur R au moyen des formules suivantes: 

Pour R < 0:  1MOS =  

Pour 0 < R < 100:  6107)100)(60(035,01MOS −⋅−−++= RRRR  (B-4) 

Pour R > 100:  5,4MOS =  

Les pourcentages de connexions GoB, PoW et la note moyenne d'opinion MOS sont donnés en 
fonction de R dans les Figures B.1 et B.2 respectivement. 

Figure B.1/G.107 −−−− Connexions GoB ("au moins bonnes") et PoW  
("au mieux médiocres") en fonction du facteur d'évaluation R 

T1207880-96

Figure B.2/G.107 −−−− La note moyenne d'opinion MOS en  
fonction du facteur d'évaluation R 

Dans certains cas, les planificateurs de la transmission peuvent ne pas être familiers avec l'utilisation 
de mesures de la qualité telles que le facteur R obtenu à partir de calculs de planification, et ainsi une 
information provisoire d'interprétation des facteurs R calculés pour les besoins en planification est 
fournie dans le Tableau B.12. Ce tableau contient également des valeurs de R transformées en MOS, 
GoB et PoW. 

____________________ 
2  L'origine du Tableau B.1 est le Tableau 1/G.109 [3].  

Tableau B.1/G.107 −−−− Information provisoire de la relation entre la valeur R  
et la satisfaction de l'usager 

Valeur R 
(limite 

inférieure) 

MOS 
(limite 

inférieure) 

GoB (%) 
(limite 

inférieure) 

PoW (%) 
(limite 

supérieure) 
Satisfaction de l'usager 

90 4,34 97 ~0 Très satisfait 
80 4,03 89 ~0 Satisfait 
70 3,60 73 6 Quelques usagers insatisfaits 
60 3,10 50 17 Beaucoup d'usagers insatisfaits 
50 2,58 27 38 Presque tous insatisfaits 

ANNEXE C 

Code source pour G.107_2 en BASIC 

 
1 CLS
2 PRINT "PROGRAM g107_2"
3 REM THIS VERSION IS CONFORM WITH THE ALGORITHM
4 REM DESCRIBED IN REC. G.107
5 REM PROGRAM WRITTEN BY N.O. JOHANNESSON
6 REM LAST MODIFICATIONS BY S. MOELLER, July 1999
7 PRINT
8 PRINT "E-model, algorithm according to ITU-T Rec. G.107 (05/00) Annex C,"
9 PRINT "for voice communication between side (S) and (R)."
10 PRINT
11 PRINT "The E-model gives an estimation of voice transmission quality"
12 PRINT "mouth-to-ear as perceived at Side (R). The model output is a"
13 PRINT "transmission rating factor R which can be transformed to give"
14 PRINT "estimates of customer opinion. Such estimates are only made for"
15 PRINT "transmission planning purposes and not for actual customer"
16 PRINT "opinion prediction. See ITU-T Rec. G.107 (05/00) for more details."
17 PRINT
18 PRINT "Note: To continue program:"
19 PRINT "a) INPUT appropriate value as indicated on screen,"
20 PRINT "or"
21 PRINT "b) if no value is asked for, PRESS THE RETURN KEY."
22 PRINT
23 DIM X(100), Y(5, 100), Col(5), Pxo(5)
24 REM Set parameters at default values
25 GOSUB 29
26 REM Goto main menu
27 GOTO 199

29 REM Default Parameter Values, according to Tab.3/G.107 (lines 29-50)
30 SLR = 8
31 RLR = 2
32 STMRs = 15
33 Ds = 3
34 STMR = 15
35 Dr = 3
36 LSTRr = 18
37 TELR = 65
38 T = 0
39 WEPL = 110
40 Tr = 0
41 Ta = 0
42 Ie = 0
43 A = 0
44 Nc = -70
45 Ps = 35
46 Pr = 35

47 qdu = 1
48 Nfor = -64
50 RETURN

198 REM Main menu (lines 198-370)
199 PRINT
200 PRINT "Print current parameter values ", "=1"
210 PRINT "Input new parameter values ", "=2"
220 PRINT "Compute table, one parameter ", "=3"
230 PRINT "Set parameter at default values", "=4"
240 PRINT "Exit program ", "=5"
250 PRINT
260 INPUT Y1
270 CLS
280 IF Y1 = 1 THEN GOSUB 500
290 IF Y1 = 2 THEN GOSUB 1000
300 IF Y1 = 3 THEN GOSUB 2000
310 IF Y1 = 4 THEN GOSUB 30
320 IF Y1 = 5 THEN GOTO 9999
330 CLS
340 IF Y1 = 4 THEN PRINT , "Parameters set at default values !"
350 GOTO 199

500 REM SUB Print current parameter values (lines 500-700)
510 PRINT , "SLR="; SLR, "RLR="; RLR, "OLR= SLR + RLR="; SLR + RLR
520 PRINT , "Side (S): Ds="; Ds
530 PRINT , "Side (R): STMR="; STMR, "Dr="; Dr, "LSTR="; STMR + Dr
540 PRINT
550 PRINT , "TELR="; TELR, "Mean One-way Delay T ms="; T
560 PRINT , "WEPL="; WEPL, "Round-trip Delay Tr ms="; Tr
570 PRINT , "One-way Absolute Delay Ta ms="; Ta
580 PRINT
590 PRINT , "Noise Floor at Side (R) Nfor dBmp="; Nfor
600 PRINT , "Circuit Noise Nc dBm0p="; Nc
610 PRINT , "Room Noise, Side (S), Ps dB(A)="; Ps
620 PRINT , "Room Noise, Side (R), Pr dB(A)="; Pr
630 PRINT
640 PRINT , "qdu="; qdu
650 PRINT
660 PRINT , "Equipment Impairment Factor Ie="; Ie
670 PRINT , "Advantage Factor A="; A
680 PRINT
690 INPUT C$
700 RETURN

1000 REM SUB Input Parameters (lines 1000-1270)
1020 CLS
1030 PRINT "Type designation of parameter for which the value is to be changed !"
1031 PRINT
1032 PRINT "Note 1. New value of OLR is obtained indirectly, i.e. by new"
1033 PRINT "value of SLR or RLR. (OLR=SLR+RLR.)"
1034 PRINT
1035 PRINT "Note 2. New value of LSTR is obtained indirectly, i.e. by new"
1036 PRINT "value of STMR or Dr. (LSTR=STMR+Dr.)"
1037 PRINT
1040 INPUT "Parameter:"; A$
1050 INPUT "New Value="; Px
1060 PRINT A$; "="; Px
1070 IF ((A$ = "SLR") OR (A$ = "slr") OR (A$ = "Slr")) THEN SLR = Px
1080 IF ((A$ = "RLR") OR (A$ = "rlr") OR (A$ = "Rlr")) THEN RLR = Px
1090 IF ((A$ = "STMR") OR (A$ = "stmr") OR (A$ = "Stmr")) THEN STMR = Px
1100 IF ((A$ = "Dr") OR (A$ = "DR") OR (A$ = "dr")) THEN Dr = Px
1110 IF ((A$ = "Ds") OR (A$ = "DS") OR (A$ = "ds")) THEN Ds = Px
1120 IF ((A$ = "TELR") OR (A$ = "telr") OR (A$ = "Telr")) THEN TELR = Px
1130 IF ((A$ = "T") OR (A$ = "t")) THEN T = Px
1140 IF ((A$ = "WEPL") OR (A$ = "wepl") OR (A$ = "Wepl")) THEN WEPL = Px
1150 IF ((A$ = "Tr") OR (A$ = "TR") OR (A$ = "tr")) THEN Tr = Px
1160 IF ((A$ = "Ta") OR (A$ = "TA") OR (A$ = "ta")) THEN Ta = Px
1170 IF ((A$ = "Ie") OR (A$ = "IE") OR (A$ = "ie")) THEN Ie = Px
1180 IF ((A$ = "A") OR (A$ = "a")) THEN A = Px

1190 IF ((A$ = "Nc") OR (A$ = "NC") OR (A$ = "nc")) THEN Nc = Px
1200 IF ((A$ = "Ps") OR (A$ = "PS") OR (A$ = "ps")) THEN Ps = Px
1210 IF ((A$ = "Pr") OR (A$ = "PR") OR (A$ = "pr")) THEN Pr = Px
1220 IF ((A$ = "qdu") OR (A$ = "QDU") OR (A$ = "Qdu")) THEN qdu = Px
1230 IF ((A$ = "Nfor") OR (A$ = "NFOR") OR (A$ = "nfor")) THEN Nfor = Px
1240 PRINT
1250 IF Y1 = 2 THEN INPUT "More parameters changed, Yes(1) or No(0)"; Ypar
1260 IF Ypar = 1 THEN GOTO 1020
1270 RETURN

2000 REM SUB Tabulate (lines 2000-3000)
2020 INPUT "Variable Parameter:"; A$
2030 PRINT "(To exit tabulation, put parameter value = 1000 !)"
2040 PRINT TAB(8); A$; TAB(18); "R"; TAB(28); "GOB %"; TAB(38); "POW %"; TAB(48);
"MOS"
2050 INPUT Px
2060 IF Px = 1000 THEN GOTO 3000
2070 IF ((A$ = "SLR") OR (A$ = "slr") OR (A$ = "Slr")) THEN SLR = Px
2080 IF ((A$ = "RLR") OR (A$ = "rlr") OR (A$ = "Rlr")) THEN RLR = Px
2090 IF ((A$ = "STMR") OR (A$ = "stmr") OR (A$ = "Stmr")) THEN
2100 STMR = Px
2110 LSTR = STMR + Dr
2120 END IF
2130 IF ((A$ = "Dr") OR (A$ = "DR") OR (A$ = "dr")) THEN
2140 Dr = Px
2150 LSTR = STMR + Dr
2160 END IF
2170 IF ((A$ = "TELR") OR (A$ = "telr") OR (A$ = "Telr")) THEN TELR = Px
2180 IF ((A$ = "T") OR (A$ = "t")) THEN T = Px
2190 IF ((A$ = "WEPL") OR (A$ = "wepl") OR (A$ = "Wepl")) THEN WEPL = Px
2200 IF ((A$ = "Tr") OR (A$ = "TR") OR (A$ = "tr")) THEN Tr = Px
2210 IF ((A$ = "Ta") OR (A$ = "TA") OR (A$ = "ta")) THEN Ta = Px
2220 IF ((A$ = "Ie") OR (A$ = "IE") OR (A$ = "ie")) THEN Ie = Px
2230 IF ((A$ = "A") OR (A$ = "a")) THEN A = Px
2240 IF ((A$ = "Nc") OR (A$ = "NC") OR (A$ = "nc")) THEN Nc = Px
2245 IF ((A$ = "Nfor") OR (A$ = "NFOR") OR (A$ = "nfor")) THEN Nfor = Px
2250 IF ((A$ = "Ps") OR (A$ = "PS") OR (A$ = "ps")) THEN Ps = Px
2260 IF ((A$ = "Pr") OR (A$ = "PR") OR (A$ = "pr")) THEN Pr = Px
2270 IF ((A$ = "qdu") OR (A$ = "QDU") OR (A$ = "Qdu")) THEN qdu = Px
2280 IF ((A$ = "Ie") OR (A$ = "IE") OR (A$ = "ie")) THEN Ie = Px
2290 IF ((A$ = "Ds") OR (A$ = "DS") OR (A$ = "ds")) THEN Ds = Px
2300 GOSUB 3500
2400 GOSUB 4000
2500 GOSUB 4100
2600 GOSUB 4200
2700 R = INT(R * 10 + .5) / 10
2800 PRINT TAB(8); Px; TAB(18); R; TAB(28); GOB; TAB(38); POW; TAB(48); MOS
2900 GOTO 2050
3000 RETURN

3500 REM Compute R (lines 3500-3880)

3509 REM Noise Summation, formulas (3) to (7)
3510 Nr1 = Ps - SLR - Ds - 100
3520 Nr1 = Nr1 + .004 * (Ps - SLR - RLR - Ds - 14) ^ 2
3530 LSTR = STMR + Dr
3540 Pro = Pr + 10 * LOG(1 + 10 ^ ((10 - LSTR) / 10)) / LOG(10)
3550 Pr1 = Pro + .008 * (Pro - 35) ^ 2
3560 Nr2 = Pr1 - 121 + RLR
3570 Nfo = Nfor + RLR
3580 No = 10 * LOG(10 ^ (Nr1 / 10) + 10 ^ (Nr2 / 10) + 10 ^ (Nc / 10) + 10 ^ (Nfo
/ 10)) / LOG(10)
3590 Nt = No - RLR

3599 REM Ro, formula (2)
3600 Ro = 15 - 1.5 * (SLR + No)

3609 REM Iolr, formulas (9) and (10)
3610 Xolr = SLR + RLR + .2 * (64 + Nt)
3620 Iolr = 20 * ((1 + (Xolr / 8) ^ 8) ^ (1 / 8) - Xolr / 8)

3629 REM Ist, formulas (11) and (12)
3630 STMRo = -10 * LOG(10 ^ (-STMR / 10) + 10 ^ (-TELR / 10) * EXP(-T / 4)) /
LOG(10)
3640 Ist = 10 * (1 + ((STMRo - 12) / 5) ^ 6) ^ (1 / 6) - 10
3650 Ist = Ist - 46 * (1 + (STMRo / 23) ^ 10) ^ (1 / 10) + 46

3659 REM Iq, formulas (13) to (17)
3660 IF qdu < 1 THEN qdu = 1
3670 Q = 37 - 15 * LOG(qdu) / LOG(10)
3680 G = 1.07 + .258 * Q + .0602 * Q ^ 2
3690 Iq = 15 * LOG(1 + 10 ^ ((Ro - 100) / 15) * 10 ^ (46 / 8.4 - G / 9) + 10 ^
(46 / 30 - G / 40)) / LOG(10)

3699 REM Is, formula (8)
3700 Isyn = Iolr + Ist + Iq

3709 REM TERV, formula (22)
3710 TERV = TELR + 6 * EXP(-.3 * T ^ 2) - 40 * LOG((1 + T / 10) / (1 + T / 150))
/ LOG(10)

3719 REM Modifications to satisfy formula (23)
3720 IF STMR < 9 THEN TERV = TERV + .5 * Ist

3729 REM Idte, formulas (19) to (21)
3730 Re = 80 + 2.5 * (TERV - 14)
3740 Roe = -1.5 * (No - RLR)
3750 Xdt = (Roe - Re) / 2
3760 Idte = Xdt + SQR(Xdt ^ 2 + 100)
3770 Idte = (Idte - 1) * (1 - EXP(-T))

3779 REM Modifications to satisfy formula (24)
3780 IF STMR > 15 THEN Idte = SQR(Idte ^ 2 + Ist ^ 2)

3789 REM Idle, formulas (25) and (26)
3790 Rle = 10.5 * (WEPL + 7) * (Tr + 1) ^ (-1 / 4)
3800 Xdl = (Ro - Rle) / 2
3810 Idle = Xdl + SQR(Xdl ^ 2 + 169)

3819 REM Idd, formulas (27) and (28)
3820 IF Ta < 100 THEN Idd = 0
3830 IF Ta = 100 THEN Idd = 0
3840 IF Ta > 100 THEN

X = (LOG(Ta / 100)) / LOG(2)
Idd = 25 * ((1 + X ^ 6) ^ (1 / 6) - 3 * (1 + (X / 3) ^ 6) ^ (1 / 6) + 2)

3850 END IF

3859 REM Id
3860 Id = Idte + Idle + Idd

3869 REM R, formula (1)
3870 R = Ro - Isyn - Id - Ie + A
3880 RETURN

4000 REM Compute GOB, formula (B.2) (lines 4000-4050)
4010 Z# = (R - 60) / 16
4020 GOSUB 5000
4030 GOB = 100 * F#
4040 GOB = INT(GOB * 10 + .5) / 10
4050 RETURN

4100 REM Compute POW, formula (B.3) (lines 4100-4150)
4110 Z# = (R - 45) / 16
4120 GOSUB 5000
4130 POW = 100 * (1 - F#)
4140 POW = INT(POW * 10 + .5) / 10
4150 RETURN

4200 REM Compute MOS, formula (B.4) (lines 4200-4260)
4210 MOS = 1 + R * .035 + R * (R - 60) * (100 - R) * 7 * 10 ^ (-6)
4220 MOS = INT(MOS * 100 + .5) / 100
4230 IF R < 0 THEN MOS = 1
4240 IF MOS < 1 THEN MOS = 1
4250 IF R > 100 THEN MOS = 4.5

4260 RETURN
5000 REM Norm Distr F(Z), formula (B.1) (lines 5000-5130)
5010 S# = 0
5020 N% = 0
5030 H# = Z#
5040 S# = S# + H#
5050 H# = H# * (-1) * (Z#) ^ 2 * (2 * N% + 1) / ((N% + 1) * 2 * (2 * N% + 3))
5060 N% = N% + 1
5070 IF ABS(H#) < 10 ^ (-6) THEN GOTO 5090
5080 GOTO 5040
5090 S# = S# / (SQR(2 * 3.14159265#))
5100 F# = .5 + S#
5110 F# = INT(F# * 10 ^ 5 + .5) / 10 ^ 5
5120 REM PRINT "Z="; Z#, "F(Z)="; F#, "N="; N%
5130 RETURN

9999 END

